17

NUMERICAL STUDIES OF VISCOUS, INCOMPRESSIBLE FLOW THROUGH AN ORIFICE FOR ARBITRARY REYNOLDS NUMBER

by Donald Greenspan

APPENDIX: PROGRAMMING ORIFICE FLOW

by M. McClellan

Computer Sciences Technical Report #20

May 1968

Numerical Studies of Viscous, Incompressible Flow through an Orifice for Arbitrary Reynolds Number*

by

Donald Greenspan

1. Introduction

A steady flow problem of interest to both engineers and mathematicians is that of a viscous, incompressible fluid through an orifice (see, e.g., references [1]-[7] and the additional references contained therein). In this paper we will develop a new numerical method for the study of such three dimensional problems under the assumption of axial symmetry. If and when the mathematical flows under consideration do in fact exist physically (which is often an open question), then the method will be practical, will be vastly more economical and accurate than any step-ahead method, and will apply with equal case to cases of both small and large Reynolds numbers. The power of the method is contained in the application of a simple smoothing process and in the structure of the difference equations, which for <u>all</u> Reynolds numbers yield diagonally dominant systems of linear algebraic equations.

---

*Funds for the computations described in this paper were made available by the Research Committee of the Graduate School of the University of Wisconsin.

2. A Basic Problem

Let us begin by considering a particular problem which contains the basic components of all axis symmetrical orifice flow problems. Our formulation proceeds as follows.

Let S_1, S_2, S_3 be three coaxial cylinders joined to form a channel with an orifice, as shown in Figure 2.1 . At the entrance and at the exit of the flow, we will assume Poiseuille conditions are valid. On the surface portions of S_1, S_2 and S_3, it will be assumed that the stream function is constant and that the normal derivative of the stream function is zero. Inside of S_1, S_2 and S_3, it will be assumed that the three dimensional Navier-Stokes equations govern the motion of the fluid. However, because of the axial symmetry, the problem can be formulated analytically as a plane problem in the following fashion. Let S , the plane polygon shown in Figure 2.1 whose vertices are A, B, C, D, E, F, G, H be positioned in the (r, x) plane as shown in Figure 2.2, where α_1, α_2, β_1, β_2, β_3, γ are all positive. Let R be the interior of S . Then one must find a pair of functions $\psi(x, r)$, $\Omega(x,r)$ which on R satisfy the Navier-Stokes equations

$$\frac{\partial^2\psi}{\partial r^2} - \frac{1}{r}\frac{\partial\psi}{\partial r} + \frac{\partial^2\psi}{\partial x^2} = -r^2\Omega \tag{2.1}$$

$$r\left(\frac{\partial^2\Omega}{\partial r^2} - \frac{1}{r}\frac{\partial\Omega}{\partial r} + \frac{\partial^2\Omega}{\partial x^2}\right) + \left(3 + \frac{1}{r^2}\right)\frac{\partial\Omega}{\partial r} = \Re\left(\frac{\partial\psi}{\partial r}\frac{\partial\Omega}{\partial x} - \frac{\partial\psi}{\partial x}\frac{\partial\Omega}{\partial r}\right) \tag{2.2}$$



FIGURE 2.1



FIGURE 2.2

and which on S satisfy the boundary conditions

(2.3) $\psi = 0 \; , \quad \frac{\partial \psi}{\partial r} = 0$; on AH

(2.4) $\psi = 1 \; , \quad \frac{\partial \psi}{\partial r} = 0$; on BC, DE, FG

(2.5) $\psi = 1 \; , \quad \frac{\partial \psi}{\partial x} = 0$; on CD, EF

(2.6)
$$\begin{cases} \psi = \left(\frac{r}{\beta_2}\right)^2 \left[2 - \left(\frac{r}{\beta_2}\right)^2\right] & ; \text{ on AB} \\ \psi = \left(\frac{r}{\beta_3}\right)^2 \left[2 - \left(\frac{r}{\beta_3}\right)^2\right] & ; \text{ on GH} \\ \Omega = 8 & ; \text{ on AB and GH .} \end{cases}$$

In (2.1)-(2.6), ψ is the stream function, Ω is related to the vorticity ξ by $\Omega = \xi/r$, and $\Re$ is the Reynolds number.

The formulation (2.1)-(2.6) was developed with the help of R. E. Meyer and is in accord with the definitions and assumptions of Goldstein [8]. The choice of Ω as a dependent variable in place of ξ was motivated by the form of (2.2), which lends itself directly to the numerical techniques to be developed. The choice of Poiseuille conditions at the ends of the channel was motivated by previous computations on a simpler problem [9], which indicated that certain other end condition choices yielded essentially the same flows as those for Poiseuille conditions.

3. Finite Difference Equations

Because a solution of boundary value problem (2.1)-(2.6) can almost never be given in closed form, we will direct our attention to developing a finite difference, computer oriented method for approximating a solution. In this section we will develop the difference equations which will be used.

For $h > 0$, let the five points (x, r), $(x+h, r)$, $(x, r+h)$, $(x-h, r)$, $(x, r-h)$ be denoted 0, 1, 2, 3, 4, respectively, as shown in Figure 3.1. We will, whenever convenient, use the notation u_k to represent the value of $u(x, y)$ at the point numbered k .

With regard to stream equation (2.1), it will be of value to approximate the second-order derivative terms by the standard five-point Laplace difference operator [10] and the first order derivative by a central difference, so that in terms of the notation in Figure 3.1, we will approximate (2.1) at (x, y) by the difference equation

$$\frac{-4\psi_0 + \psi_1 + \psi_2 + \psi_3 + \psi_4}{h^2} - \frac{1}{r}\left(\frac{\psi_2 - \psi_4}{2h}\right) = -r^2\Omega_0 ,$$

or, equivalently, by

$$(3.1) \quad -4\psi_0 + \psi_1 + \psi_2\left(1 - \frac{h}{2r}\right) + \psi_3 + \psi_4\left(1 + \frac{h}{2r}\right) = -r^2 h^2 \Omega_0 .$$

With regard to (2.2), one must be more subtle in order to maintain the dominance of the Ω_0 term. This will be accomplished by combining central, forward and backward differences in the following way. First rewrite (2.2) in the form

$$(3.2) \quad r\left(\frac{\partial^2\Omega}{\partial r^2} - \frac{1}{r}\frac{\partial\Omega}{\partial r} + \frac{\partial^2\Omega}{\partial x^2}\right) - \Re\,\frac{\partial\psi}{\partial r}\frac{\partial\Omega}{\partial x} + \left(3 + \frac{1}{r^2} + \Re\,\frac{\partial\psi}{\partial x}\right)\frac{\partial\Omega}{\partial r} = 0 .$$

As in the development of (3.1), we will use the approximation

$$(3.3) \quad \frac{\partial^2\Omega}{\partial r^2} - \frac{1}{r}\frac{\partial\Omega}{\partial r} + \frac{\partial^2\Omega}{\partial x^2} \sim \frac{1}{h^2}\left[-4\Omega_0 + \Omega_1 + \Omega_2\left(1 - \frac{h}{2r}\right) + \Omega_3 + \Omega_4\left(1 + \frac{h}{2r}\right)\right].$$

For $\frac{\partial\psi}{\partial r}$ and $\frac{\partial\psi}{\partial x}$, it will be convenient to use the central difference approximations

$$(3.4) \quad \frac{\partial\psi}{\partial r} = \frac{\psi_2 - \psi_4}{2h}, \quad \frac{\partial\psi}{\partial x} = \frac{\psi_1 - \psi_3}{2h} .$$

If one then sets

$$(3.5) \quad M = \Re\left(\frac{\psi_2 - \psi_4}{2h}\right), \quad N = 3 + \frac{1}{r^2} + \Re\left(\frac{\psi_1 - \psi_3}{2h}\right),$$

then $\frac{\partial\Omega}{\partial x}$ and $\frac{\partial\Omega}{\partial r}$ will be approximated as follows:

(a) if $M \geq 0$ and $N \geq 0$, set $\frac{\partial\Omega}{\partial x} = \frac{\Omega_0 - \Omega_3}{h}$ and $\frac{\partial\Omega}{\partial r} = \frac{\Omega_2 - \Omega_0}{h}$;

(b) if $M < 0$ and $N \geq 0$, set $\frac{\partial\Omega}{\partial x} = \frac{\Omega_1 - \Omega_0}{h}$ and $\frac{\partial\Omega}{\partial r} = \frac{\Omega_2 - \Omega_0}{h}$;

(c) if $M \geq 0$ and $N < 0$, set $\frac{\partial\Omega}{\partial x} = \frac{\Omega_0 - \Omega_3}{h}$ and $\frac{\partial\Omega}{\partial r} = \frac{\Omega_0 - \Omega_4}{h}$;

(d) if $M < 0$ and $N < 0$, set $\frac{\partial\Omega}{\partial x} = \frac{\Omega_1 - \Omega_0}{h}$ and $\frac{\partial\Omega}{\partial r} = \frac{\Omega_0 - \Omega_4}{h}$.

Thus, making the above substitutions into (3.2) yields readily the following



FIGURE 3.1



FIGURE 3.2

difference equation approximations for (3.2) :

(3.6a) $$\Omega_0 \left(-4 - \frac{Mh}{r} - \frac{Nh}{r}\right) + \Omega_1 + \Omega_2 \left(1 - \frac{h}{2r} + \frac{Nh}{r}\right) + \Omega_3 \left(1 + \frac{Mh}{r}\right) + \Omega_4 \left(1 + \frac{h}{2r}\right) = 0, \quad \text{if} \quad M \geq 0, \; N \geq 0 ;$$

(3.6b) $$\Omega_0 \left(-4 + \frac{Mh}{r} - \frac{Nh}{r}\right) + \Omega_1 \left(1 - \frac{Mh}{r}\right) + \Omega_2 \left(1 - \frac{h}{2r} + \frac{Nh}{r}\right) + \Omega_3 + \Omega_4 \left(1 + \frac{h}{2r}\right) = 0 , \quad \text{if} \quad M < 0, \; N \geq 0 ;$$

(3.6c) $$\Omega_0 \left(-4 - \frac{Mh}{r} + \frac{Nh}{r}\right) + \Omega_1 + \Omega_2 \left(1 - \frac{h}{2r}\right) + \Omega_3 \left(1 + \frac{Mh}{r}\right) + \Omega_4 \left(1 + \frac{h}{2r} - \frac{Nh}{r}\right) = 0 , \quad \text{if} \quad M \geq 0, \; N < 0 ;$$

(3.6d) $$\Omega_0 \left(-4 + \frac{Mh}{r} + \frac{Nh}{r}\right) + \Omega_1 \left(1 - \frac{Mh}{r}\right) + \Omega_2 \left(1 - \frac{h}{2r}\right) + \Omega_3 + \Omega_4 \left(1 + \frac{h}{2r} - \frac{Nh}{r}\right) = 0 , \quad \text{if} \quad M < 0, \; N < 0 .$$

Finally, we will develop simple difference equations by means of which Ω can be approximated at points of S which are not in AB or GH. Consider first the four points (x, r), $(x+h, r)$, $(x, r+h)$, $(x, r-h)$, numbered 0, 1, 2, 4, respectively, in Figure 3.2(a). Let us try to determine parameters α_0, α_1, α_2, α_4, α_5 such that

(3.7) $$\left(\frac{\partial^2 \psi}{\partial r^2} - \frac{1}{r}\frac{\partial \psi}{\partial r} + \frac{\partial^2 \psi}{\partial x^2}\right)\Bigg|_0 \equiv \alpha_0 \psi_0 + \alpha_1 \psi_1 + \alpha_2 \psi_2 + \alpha_4 \psi_4 + \alpha_5 \left(\frac{\partial \psi}{\partial x}\right)\Bigg|_0 .$$

In (3.7), expansion of ψ_1, ψ_2 and ψ_4 in Taylor series implies

$$
\begin{aligned}
(\psi_{rr} - \frac{1}{r}\,\psi_r + \psi_{xx}) \equiv\ & \psi_0\,(\alpha_0 + \alpha_1 + \alpha_2 + \alpha_4) \\
& + \psi_x\,(h\alpha_1 + \alpha_5) \\
& + \psi_r\,(h\alpha_2 - h\alpha_4) \\
& + \psi_{xx}(\frac{h^2}{2}\,\alpha_1) \\
& + \psi_{rr}(\frac{h^2}{2}\,\alpha_2 + \frac{h^2}{2}\,\alpha_4) \\
& + \ldots\ .
\end{aligned}
$$

In this latter identity, setting corresponding coefficients equal yields

$$
\begin{aligned}
\alpha_0 + \alpha_1 + \alpha_2 + \alpha_4 &= 0 \\
h\alpha_1 + \alpha_5 &= 0 \\
h\alpha_2 - h\alpha_4 &= -\frac{1}{r} \\
\frac{h^2}{2}\,\alpha_1 &= 1 \\
\frac{h^2}{2}\,\alpha_2 + \frac{h^2}{2}\,\alpha_4 &= 1\ ,
\end{aligned}
$$

the solution of which is

$$
\alpha_0 = -\frac{4}{h^2}\,,\quad \alpha_1 = \frac{2}{h^2}\,,\quad \alpha_2 = \frac{1}{h^2} - \frac{1}{2hr}\,,\quad \alpha_4 = \frac{1}{h^2} + \frac{1}{2hr}\,,\quad \alpha_5 = -\frac{2}{h}\ .
$$

Thus one has the following approximation :

$$
\begin{aligned}
\frac{\partial^2\psi}{\partial r^2} - \frac{1}{r}\,\frac{\partial\psi}{\partial r} + \frac{\partial^2\psi}{\partial x^2} = & -\frac{4}{h^2}\,\psi_0 + \frac{2}{h^2}\,\psi_1 + (\frac{1}{h^2} - \frac{1}{2hr})\,\psi_2 \\
& + (\frac{1}{h^2} + \frac{1}{2hr})\,\psi_4 - \frac{2}{h}\left(\frac{\partial\psi}{\partial x}\right)\Big|_0\ .
\end{aligned}
$$

Similarly, for the four points (x, r), $(x+h, r)$, $(x, r+h)$, $(x-h, r)$, numbered 0, 1, 2, 3, respectively, in Figure 3.2(b), one has

$$(3.8b)\qquad \psi_{rr} - \frac{1}{r}\psi_r + \psi_{xx} = -\frac{4}{h^2}\psi_0 + \frac{1}{h^2}\psi_1 + \frac{2}{h^2}\psi_2 + \frac{1}{h^2}\psi_3 - \left(\frac{1}{r} + \frac{2}{h}\right)(\psi_r)\Big|_0 ;$$

for the four points (x, r), $(x, r+h)$, $(x-h, r)$, $(x, r-h)$, numbered 0, 2, 3, 4, respectively, in Figure 3.2(c), one has

$$(3.8c)\qquad \psi_{rr} - \frac{1}{r}\psi_r + \psi_{xx} = -\frac{4}{h^2}\psi_0 + \left(\frac{1}{h^2} - \frac{1}{2hr}\right)\psi_2 + \frac{2}{h^2}\psi_3 + \left(\frac{1}{h^2} + \frac{1}{2hr}\right)\psi_4 + \frac{2}{h}(\psi_x)\Big|_0 ;$$

and for the four points (x, r), $(x+h, r)$, $(x-h, r)$, $(x, r-h)$, numbered 0, 1, 3, 4, respectively, in Figure 3.2(d), one has

$$(3.8d)\qquad \psi_{rr} - \frac{1}{r}\psi_r + \psi_{xx} = -\frac{4}{h^2}\psi_0 + \frac{1}{h^2}\psi_1 + \frac{1}{h^2}\psi_3 + \frac{2}{h^2}\psi_4 + \left(\frac{2}{h} - \frac{1}{r}\right)(\psi_r)\Big|_0 .$$

Note that the numbering of the points in Figure 3.2 is consistent with that in Figure 3.1 .

4. <u>The Numerical Method</u>

Forfixed positive integer n, determine grid size $h = \frac{1}{n}$. Next, on and within polygon A B C D E F G H (see Figure 2.2), construct and number in the usual way [10] the set of interior grid points R_h and the set of boundary grid points S_h. With regard to R_h and S_h, it will be assumed, with very little loss of generality, that h can be selected so that α_1, α_2, β_1, β_2, β_3, and γ are integral multiples of h .

We will aim at constructing on $R_h + S_h$ a pair of finite sequences of discrete functions

(4.1) $\psi^{(0)}, \psi^{(1)}, \psi^{(2)}, \ldots, \psi^{(k)}, \psi^{(k+1)}$

(4.2) $\Omega^{(0)}, \Omega^{(1)}, \Omega^{(2)}, \ldots, \Omega^{(k)}, \Omega^{(k+1)}$

with the properties that for some given tolerance ε

(4.3) $|\psi^{(k)} - \psi^{(k+1)}| < \varepsilon$

(4.4) $|\Omega^{(k)} - \Omega^{(k+1)}| < \varepsilon$

at each point of $R_h + S_h$. Each function in sequence (4.1) and in sequence (4.2) will be called an outer iterate and the particular functions $\psi^{(k)}$ and $\Omega^{(k)}$ will be taken to be approximations on $R_h + S_h$ to $\psi(x, y)$ and $\Omega(x, y)$, respectively.

For the above purpose, we begin by defining $\psi^{(0)}$ and $\Omega^{(0)}$ as follows. At each grid point in AH set $\psi^{(0)} = 0$. At each grid point in BC, CD, DE, EF, and FG, set $\psi^{(0)} = 1$. At each grid point in AB and GH, determine $\psi^{(0)}$ from (2.6). And on the remaining grid points in $R_h + S_h$, determine $\psi^{(0)}$ by linear interpolation along vertical grid lines. At each point in AB and GH, let $\Omega^{(0)} = 8$, and on the remainder of $R_h + S_h$ let $\Omega^{(0)} = 0$.

The second element of sequence (4.1) is now determined as follows. At each grid point in AH set $\psi^{(1)} = 0$. At each grid point in BC, CD, DE, EF, and FG, set $\psi^{(1)} = 1$. At each grid point in AB and GH, determine $\psi^{(1)}$ from (2.6). Next, at each grid point (x, r) in R_h, write down (3.1) with $\Omega_0 = \Omega_0^{(0)}$. The resulting system of linear algebraic equations is then solved

by the generalized Newton's method [10] with over-relaxation factor r_ψ. The solution is denoted by $\overline{\psi}^{(1)}$, which is of course defined only on R_h. The function $\psi^{(1)}$ is then determined on R_h by the smoothing formula

(4.5) $$\psi^{(1)} = \rho\,\psi^{(0)} + (1-\rho)\,\overline{\psi}^{(1)} \quad , \quad 0 \le \rho \le 1 ,$$

thus completing the definition of $\psi^{(1)}$ on all of $R_h + S_h$.

The second element of sequence (4.2) is determined as follows. At each grid point in AB and GH, let $\Omega^{(1)} = 8$. At each grid point interior to BC, interior to DE and interior to FG, use (3.8d) in the form

(4.6) $$\overline{\Omega}_0^{(1)} = \frac{2}{r^2h^2}\,[1 - \psi_4^{(1)}]$$

to approximate Ω_0. At each grid point interior to CD, use (3.8c) in the form

(4.7) $$\overline{\Omega}_0^{(1)} = \frac{2}{r^2h^2}\,[1 - \psi_3^{(1)}]$$

to approximate Ω_0. At each grid point interior to EF, use (3.8a) in the form

(4.8) $$\overline{\Omega}_0^{(1)} = \frac{2}{r^2h^2}\,[1 - \psi_1^{(1)}]$$

to approximate Ω_0. At C and at F set $\Omega_0^{(1)} = 0$. At D approximate Ω_0 by

(4.9) $$\overline{\Omega}_0^{(1)} = \frac{1}{r^2h^2}\,[2 + \frac{h}{2r} - \psi_3^{(1)} - \psi_4^{(1)}(1 + \frac{h}{2r})]$$

and at E approximate Ω_0 by

(4.10) $$\overline{\Omega}_0^{(1)} = \frac{1}{r^2h^2}\,[2 + \frac{h}{2r} - \psi_1^{(1)} - \psi_4^{(1)}(1 + \frac{h}{2r})] .$$

Next, on R_h and at each point of S_h interior to AH determine a system of algebraic equations as follows. At each point of S_h interior to AH, set

$$\Omega(x, 0) = \Omega(x, h) \quad , \tag{4.11}$$

and at each point of R_h, write down the appropriate equation from (3.6a)-(3.6d), where $\psi^{(1)}$ is used for ψ. Then, with over-relaxation factor r_Ω, solve the resulting system and call the solution $\overline{\Omega}^{(1)}$.

Finally, determine $\Omega^{(1)}$ at those points of $R_h + S_h$ where only $\overline{\Omega}^{(0)}$ has been defined by the smoothing formula

$$\Omega^{(1)} = \mu\Omega^{(0)} + (1 - \mu)\overline{\Omega}^{(1)} \, , \quad 0 \le \mu \le 1 \, , \tag{4.12}$$

thus completing the definition of $\Omega^{(1)}$ on all of $R_h + S_h$.

The numerical method then proceeds by generating $\psi^{(2)}$ from $\Omega^{(1)}$ just as $\psi^{(1)}$ was generated from $\Omega^{(0)}$ and by generating $\Omega^{(2)}$ from $\psi^{(2)}$ just as $\Omega^{(1)}$ was generated from $\psi^{(1)}$. The indicated iteration continues until, for some k, (4.3) and (4.4) are valid. Substitution of $\psi^{(k)}$ and $\Omega^{(k)}$ into the difference approximations of (2.1) and (2.2), to assure that these are the solutions, terminates the method.

5. Examples.

We will try now to organize in some comprehensive way the large number of examples run on the CDC 3600 at the University of Wisconsin.

Convergent results were obtained easily for $10 \le \Re < 500$ with $\beta_1 = \gamma = \frac{1}{2}$, $\beta_2 = \beta_3 = 1$, $r_\psi = 1.8$, $r_\Omega = 1.0$, $\varepsilon = 10^{-4}$ and $h = \frac{1}{10}$. Unfortunately, the present choices of ε and h were dictated by economic considerations. The resulting flows are all described qualitatively in Figure 5.1 . Quantitative aspects of the computations are given in Table 5.1 . Note also that from time to time the channel was doubled in length to check on the accuracy of the results.

Table 5.1

$\Re$	α_1	α_2	ρ	μ	Length of Vortex at F	Max ψ and point where it occurs	No. outer iterations	Running time
10	4	4	0.1	0.7	1.3	1.061, (0.9, 0.7)	40	8 min 8 sec
50	15	4	0.1	0.8	6.7	1.109, (1.8, 0.7)	70	22 min 18 sec
100	25	4	0.1	0.8	13.5	1.119, (2.8, 0.7)	70	40 min 5 sec
200	30	4	0.1	0.8	26.8	1.125, (4.9, 0.7)	70	58 min 27 sec

It was hypothesized from the results in Table 5.3 that the length ℓ of the vortex at F varied directly with the Reynolds number $\Re$ by the relationship

$$\ell = 0.134\ \Re \ .$$

The vortex at F in Figure 5.2 became so large for $\Re = 500$ that the channel had to be modified in order to study flows for higher Reynolds numbers. Thus, for $\Re$ = 500, 1000, 2000, 3000, 5000, 10000, and 25000, S_3 was eliminated in Figure 2.1 by setting $\alpha_1 = \gamma$. The resulting two dimensional



FIGURE 5.1



FIGURE 5.2



FIGURE 5.3

configuration is shown in Figure 5.2. The compuations proceeded with $\alpha_2 = 7.5$, $\gamma = 2.5$, $\beta_2 = \beta_3 = 1$, $\beta_1 = \frac{1}{2}$, $r_\psi = 1.8$, $r_\Omega = 1.0$, $\varepsilon = 10^{-4}$, $h = \frac{1}{10}$, and with Poiseuille conditions

$$\psi = 8r^2 (1 - 2r^2), \quad \Omega = 8$$

at exit EJ in Figure 5.2. For each Reynolds number convergence was achieved in fewer than 60 iterations and in less than 6 minutes of running time. For $\Re = 500$ no vortex appeared at C, but for each successive choice of $\Re$ a small vortex did appear, as shown in Figure 5.2. And although the length of this vortex increased with $\Re$, it was still no greater in length than 0.01 for $\Re = 25000$.

For still another type of channel, the problem for $\Re = 10$, which was described in Table 5.1, was modified by the insertion of a 45° wedge, as shown in Figure 5.3, the coordinates of K being $(0, \frac{1}{2})$. The method of Section 4 had to be modified only in its approximation of Ω at points of CK and KF as follows. At K, let $(0, \frac{1}{2})$, $(h, \frac{1}{2})$, $(-h, \frac{1}{2})$, $(0, \frac{1}{2} - h)$ be numbered 0, 1, 3, 4, respectively, as shown in Figure 5.4a, and approximate Ω at K by

$$(5.1) \qquad \overline{\Omega}_0 = \frac{4}{h^2} (4 - \psi_1 - \psi_3 - 2\psi_4) .$$

If (x, r) is a grid point between C and K, let (x, r), $(x-h, r+h)$, $(x-h, r-h)$, $(x+h, r-h)$ be numbered 0, 6, 7, 8, respectively, as shown in Figure 5.4b, and approximate Ω at (x, r) by

$$(5.2) \qquad \overline{\Omega}_0 = \frac{1}{r^2h^2} (1 - \psi_7) .$$



FIGURE 5.4

If (x, r) is a grid point between K and F, let (x, r), (x+h, r+h), (x-h, r-h), (x+h, r-h) be numbered 0, 5, 7, 8, as shown in Figure 5.4c, and approximate Ω at (x, r) by

(5.3) $$\overline{\Omega}_0 = \frac{1}{r^2h^2}(1 - \psi_8) .$$

If C is numbered 0, let $(-\frac{1}{2} - h, 1)$, $(-\frac{1}{2}, 1-h)$, $(-\frac{1}{2} - h, 1-h)$, $(-\frac{1}{2} + h, 1-h)$ be numbered 3, 4, 7, 8, respectively, as shown in Figure 5.4d, and approximate Ω at C by

(5.4) $$\overline{\Omega}_0 = \frac{2}{h^2}(1 - \psi_4) .$$

Finally, if F is numbered 0 and $(\frac{1}{2} + h, 1)$, $(\frac{1}{2}, 1-h)$, $(\frac{1}{2} - h, 1-h)$, $(\frac{1}{2} + h, 1-h)$ are numbered 1, 4, 7, 8, respectively, as shown in Figure 5.4e, then approximate Ω at F by

(5.5) $$\overline{\Omega}_0 = \frac{2}{h^2}(1 - \psi_4) .$$

The derivatives of (5.1)-(5.5) are completely analogous to those for (4.6)-(4.10). The resulting flow is shown qualitatively in Figure 5.3. The vortex extended to $x = 1.5$ and the maximum value of ψ was $\psi(0.6, 0.7) = 1.064$. Convergence was achieved in 40 outer iterations and 9 minutes 49 seconds of running time.

With those computingfunds which remained, it was decided to repeat one of the previously run problems but with a refined grid. The problem selected was that described for $\mathcal{R} = 10$ in Table 5.1. The only parameters which were changed were $h = \frac{1}{20}$, $\rho = 0.05$, $\mu = 0.85$ and $\varepsilon = 10^{-6}$.



$\psi = 1.4$
$\psi = 1.2$
$\psi = 1.0005$
(-1,1)
C
O
F
(2.5,1)
$\psi = 0.90$
$\psi = 0.75$
$\psi = 0.50$
$\psi = 0.10$
D
E
(-1,0)
(2.5,0)

FIGURE 5.4

Convergence resulted in 116 outer iterations and 97 minutes of running time. The resulting flow is shown between $x = -1$ and $x = 2.5$ in Figure 5.4. The small vortex at C is centered at (-.55, .95) and has a length smaller than 0.05. The large vortex at F is 1.35 units long and has a maximum ψ value of 1.0529 at (0.95, 0.75). It is also of interest to note that the same streamlines result for both $\varepsilon = 10^{-4}$ and $\varepsilon = 10^{-6}$, because the $\psi^{(n)}$ sequence converges much more rapidly than the $\Omega^{(n)}$ sequence. Finally, we observe that the construction of the small vortex at C implies that it existed in all previously considered Figure 5.1 and Figure 5.2 type problems but that a grid size of $h = \frac{1}{10}$ was simply too coarse to detect it.

6. Remark

Because the work in this paper is experimental in nature, it is necessary that other workers in the field be able to reproduce our examples in order to support or to refute our results. For this purpose, we are including in an appendix the complete Fortran program used to calculate the flows for the channel shown in Figure 5.1 .

References

1. M. Bogema, S. Bradford and M. V. Ramamoorthy, "Quadrant edge orifice performance-effect of upstream velocity distribution," Trans. ASME, Paper 61-WA-28, 1961.

2. B. W. Hunt, "Numerical solution of an integral equation for flow from a circular orifice," J. Fluid Mech., 31, 1968, pp. 361-378.

3. H. W. Liepmann, "Gas kinetics and gas dynamics of orifice flow," J. Fluid Mech., 10, 1961, pp. 65-79.

4. P. L. Monkmeyer, "Studies on the quadrant edge orifice," Thesis, Cornell University, Ithaca, N. Y., 1956.

5. R. Narasimha, "Orifice flow at high Knudsen numbers," J. Fluid Mech., 10, 1961, pp. 371-384.

6. I. L. Parmet and E. Saibel, "Axisymmetric creeping flow from an orifice in a plane wall," Comm. Pure Appl. Math., 18, 1965, pp. 17-23.

7. L. Prandtl and O. Tietjens, Applied Hydro- and Aeromechanics, Engineering Societies Monograph, McGraw-Hill Book Co., New York, 1934.

8. S. Goldstein, Modern Developments in Fluid Dynamics, Oxford University Press, London, 1938.

9. D. Greenspan, "Numerical studies of steady, viscous, incompressible flow in a channel with a step", Tech. Rpt. 17, Dept. Computer Sciences, Univ. of Wisconsin, 1968.

10. D. Greenspan, Introductory Numerical Analysis of Elliptic Boundary Value Problems, Harper and Row, N. Y., 1965.

APPENDIX

Definitions of Main Program Variables and Parameters.

PSI, W = stream function and vorticity vectors.

XL ≤ XSL ≤ XSR ≤ XR = left end of channel, left end of orifice right end of orifice, right end of channel, resp.

D = width of channel at orifice (D < 1,0, max width)

H = grid size

M, N = number of vertical and horizontal lines in grid, resp.

OMEGAP = relaxation factor for PSI inner-iterations

OMEGAW = relaxation factor for W inner-iterations

XI, DELTA = weights for PSI and W, resp.

TOLP = tolerance for PSI inner- and outer-iterations.

TOLW = tolerance for W inner- and outer-iterations.

TOLTEST = number of outer-iterations between tests for problem convergence.

TOLTESTP = number of PSI inner-iterations between tests for convergence.

TOLTESTW = number of W inner-iterations between tests for convergence.

ITERMAX = maximum number of outer-iterations for both PSI and W.

ITERMAXP = maximum number of inner-iterations for PSI.

ITERMAXW = maximum number of inner-iterations for W.

WRTP = number of outer-iterations between commands to write stream function and vorticity vectors on tape. Must be multiple of TOLTEST.

Program switches

PCONV = 1, if PSI outer-iterations have converged.

TAPEUSE = 1, if PSI and W are to be saved on tape.

INPTAPE = 1, if PSI and W are to be initialized from tape.

RET = return address from W relaxation-coefficients block.

```
      PROGRAM NS5
      DIMENSION PSI(161,21),W(161,21),PSISAV(161,21),WSAV(161,21)
      COMMON/PARAMS/ M,N,M1,MSL,MSR,MR,MSLP1,NB,NST,MP,NP,XSL,XSR,XL,
     1 H,MS,D
      TYPE INTEGER TOLTEST,TOLTESTP,TOLTESTW,RET,WRTP
      TYPE LOGICAL PCONV,TAPEUSE,INPTAPE
C
C    READ INPUT.
      READ 901, NPROBS
  901 FORMAT(I5)
      DO 70 IPROB =1,NPROBS
      READ 904, XL,XSL,XSR,XR,D,H,OMEGAP,OMEGAW,XI,DELTA,TOLP,TOLW,R
  904 FORMAT(10F5,2E5,E10)
      READ 906, MP,NP,ITERMAX,TOLTEST,ITERMAXP,TOLTESTP,ITERMAXW,
     1 TOLTESTW,TAPEUSE,INPTAPE,WRTP
  906 FORMAT(8I5,2L5,I5)
C
C    COMPUTE INITIAL PARAMETERS.
      XCH=XR-XL
      XCHL=XSL-XL
      XCHR=XR-XSR
      XS=XSR-XSL
      M=XCH/H+1.5
      MSL=XCHL/H+1.5
      MSR=(XSR-XL)/H+1.5
      MS=XS/H+1.5
      MR=XCHR/H+1.5
      MSLP1=MSL+1
      MSL1=MSL-1
      MSRP1=MSR+1
      MSR1=MSR-1
      N=1.0/H+1.5
      NB=D/H+1.5
      NBP1=NB+1
      NB1=NB-1
      NST=NB
      M1=M-1
      N1=N-1
      NPTS=M*N-(MS-2)*(N-NB)
      H2=H*H
      H4=H2*H2
      CP00=1.-OMEGAP
      CP0=0.25*OMEGAP
      CW00=1.0-OMEGAW
      CW5=2./H2
      CW6=1./(D*D*H2)
      CW7=.5*H/D
      CW8=2.*CW6
      CW9=2./H4
      CW10=.5*R/H
      CW11=1./H2
      R2=0.5*R
      CW=1.0
      XI1=1.-XI
      DELTA1=1.-DELTA
      RMAX=RMAXPSV=1.E+5
```

```
      ITER=ITOL=0
C
C     PRINT INITIAL PARAMETERS.
      PRINT 909, XCH,XCHL,XCHR,XS,CW,D,H,N,M,NPTS,OMEGAP,OMEGAW,
     1 TOLP,TOLW,TOLTEST,ITERMAX
 909  FORMAT(1H1,10X,13HPROBLEM NO. 5,10X,51HNAVIER-STOKES EQUATIONS FOR
     1 FLOW THROUGH AN ORIFICE //
     2 20X,19HLENGTH OF CHANNEL =,F8.4,31H  WITH LENGTH TO LEFT OF STEP
     2=,F8.4 / 49X,29HAND LENGTH TO RIGHT OF STEP =,F8.4 /
     2 58X,20HAND LENGTH OF STEP =,F8.4 /
     3 20X,22HWIDE PART OF CHANNEL =,F8.4 /
     4 20X,24HNARROW PART OF CHANNEL =,F8.4 /
     4 20X,15HGRID SIZE (H) =,F10.6 /
     5 20X,25HNO. OF HOROZONTAL LINES =,I5,28H AND NO. OF VERTICAL LINES
     6 =,I5  / 20X,19HTOTAL GRID POINTS =,I7 /
     7 20X,39HRELAXATION FACTOR FOR STREAM FUNCTION =, F6.2 /
     7 20X,33HRELAXATION FACTOR FOR VORTICITY =,F6.2 /
     8 20X,19HTOLERANCE FOR PSI =,E10.1 / 20X,17HTOLERANCE FOR W =,E10.1
     8 / 20X,43HTOLERANCE TEST CYCLE FOR OUTER-ITERATIONS =I5 /
     9 20X,30HMAXIMUM NUMBER OF ITERATIONS =,I6 )
      PRINT 9091, ITERMAXP,TOLTESTP,ITERMAXW,TOLTESTW,R,XI,DELTA
 9091 FORMAT(20X,43HMAXIMUM ITERATIONS FOR PSI-INNER-ITERATIONS,I6 /
     1 20X,47HTOLERANCE TEST CYCLE FOR PSI INNER-ITERATIONS =,I6 /
     2    20X,41HMAXIMUM ITERATIONS FOR W-INNER-ITERATIONS,I6 /
     3 20X,45HTOLERANCE TEST CYCLE FOR W INNER-ITERATIONS =,I6 /
     4 20X,17HREYNOLDS NUMBER =,F10.2 /
     5 20X,25HWEIGHTING (XI ) FOR PSI =F6.2 /
     6 20X,25HWEIGHTING (DELTA) FOR W =,F6.2 //)
C
C     INITIALIZE VECTORS W AND PSI.
      IF(INPTAPE) 5,8
C     INITIALIZE FROM INPUT TAPE.
 5    REWIND 5
      DO 6 J=1,N
 6    READ (5) (PSI(I,J),I=1,M)
      DO 7 J=1,N
 7    READ (5) (W(I,J),I=1,M)
      GO TO 9
C     INITIALIZE BY STANDARD PROCEDURE.
 8    CALL INIT5(W,PSI)
C     PRINT INITIAL VECTORS W AND PSI.
 9    PRINT 911, ITER,RMAX
 911  FORMAT(///10X,16HAT ITERATION NO.,I6,20H  MAXIMUM RESIDUAL =,E12.4
     1 /20X,15HSTREAM FUNCTION)
      CALL PRTMAT(PSI)
      PRINT 912
 912  FORMAT(//20X,9HVORTICITY )
      CALL PRTMAT(W)
C
C
C     BEGIN MAIN LOOP.
C
C     TEST IF VECTORS TO BE SAVED ON TAPE.
 10   IF(TAPEUSE .AND. ITER.NE.0) 1003,101
 1003 P1=(ITER+0)/WRTP
      P2=(ITER+0.)/WRTP
```

```
      IF(P1 .NE. P2) GO TO 101
      REWIND 5
      DO 1005 J=1,N
 1005 WRITE (5) (PSI(I,J),I=1,M)
      DO 1006 J=1,N
 1006 WRITE (5) (W(I,J),I=1,M)
C
 101  ITER=ITER+1
      ITOL=ITOL+1
      DO 102 J=2,N1
      DO 102 I=2,M
 102  PSISAV(I,J)=PSI(I,J)
 105  RMAXP=1.E94
      ITERP=0
 106  ITOLP=0
 12   ITERP=ITERP+1
      ITOLP=ITOLP+1
      IF(ITOLP .LT. TOLTESTP) 15,20
C
C    SWEEP STREAM FUNCTION IN REGION BELOW STEP.
 15   DO 16 J=2,NB1
      CPJ=J-1
      CPJ1=2.*CPJ
      CPJ2=CPJ*CPJ
      CP2=(CPJ1-1.)/CPJ1
      CP4=(CPJ1+1.)/CPJ1
      DO 16 I=2,M1
 16   PSI(I,J)=CP00*PSI(I,J)+CP0*(PSI(I+1,J)+CP2*PSI(I,J+1)+PSI(I-1,J)
     1 +CP4*PSI(I,J-1)+CPJ2*H4*W(I,J))
C    SWEEP STREAM FUNCTION IN UPPER REGIONS TO LEFT AND RIGHT OF STEP.
      DO 18 J=NB,N1
      CPJ=J-1
      CPJ1=2.*CPJ
      CPJ2=CPJ*CPJ
      CP2=(CPJ1-1.)/CPJ1
      CP4=(CPJ1+1.)/CPJ1
      DO 17 I=2,MSL1
 17   PSI(I,J)=CP00*PSI(I,J)+CP0*(PSI(I+1,J)+CP2*PSI(I,J+1)+PSI(I-1,J)
     1 +CP4*PSI(I,J-1)+CPJ2*H4*W(I,J))
      DO 18 I=MSRP1,M1
 18   PSI(I,J)=CP00*PSI(I,J)+CP0*(PSI(I+1,J)+CP2*PSI(I,J+1)+PSI(I-1,J)
     1 +CP4*PSI(I,J-1)+CPJ2*H4*W(I,J))
      GO TO 12
C
 20   RMAX1P=0.0
C    SWEEP STREAM FUNCTION IN REGION BELOW STEP AND COMPUTE RESIDUALS.
      DO 22 J=2,NB1
      CPJ=J-1
      CPJ1=2.*CPJ
      CPJ2=CPJ*CPJ
      CP2=(CPJ1-1.)/CPJ1
      CP4=(CPJ1+1.)/CPJ1
      DO 22 I=2,M1
      PSIOLD=PSI(I,J)
      PSINEW   =CP00*PSI(I,J)+CP0*(PSI(I+1,J)+CP2*PSI(I,J+1)+PSI(I-1,J)
     1 +CP4*PSI(I,J-1)+CPJ2*H4*W(I,J))
```

```
      PSI(I,J)=PSINEW
      RES=ABSF(PSINEW-PSIOLD)
      IF(RES .GT. RMAX1P) 215,22
 215  RMAX1P=RES
 22   CONTINUE
C   SWEEP STREAM FUNCTION IN UPPER REGIONS TO LEFT AND RIGHT OF STEP,
C   AND COMPUTE RESIDUALS.
      DO 25 J=NB,N1
      CPJ=J-1
      CPJ1=2.*CPJ
      CPJ2=CPJ*CPJ
      CP2=(CPJ1-1.)/CPJ1
      CP4=(CPJ1+1.)/CPJ1
      DO 24 I=2,MSL1
      PSIOLD=PSI(I,J)
      PSINEW  =CP00*PSI(I,J)+CP0*(PSI(I+1,J)+CP2*PSI(I,J+1)+PSI(I-1,J)
     1 +CP4*PSI(I,J-1)+CPJ2*H4*W(I,J))
      PSI(I,J)=PSINEW
      RES=ABSF(PSINEW-PSIOLD)
      IF(RES .GT. RMAX1P) 235,24
 235  RMAX1P=RES
 24   CONTINUE
      DO 25 I=MSRP1,M1
      PSIOLD=PSI(I,J)
      PSINEW  =CP00*PSI(I,J)+CP0*(PSI(I+1,J)+CP2*PSI(I,J+1)+PSI(I-1,J)
     1 +CP4*PSI(I,J-1)+CPJ2*H4*W(I,J))
      PSI(I,J)=PSINEW
      RES=ABSF(PSINEW-PSIOLD)
      IF(RES .GT. RMAX1P) 245,25
 245  RMAX1P=RES
 25   CONTINUE
      RMAXP=RMAX1P
C
C   TEST PSI INNER-ITERATIONS FOR DIVERGENCE.
      IF(RMAXP.GT. 1.E+5 ) 32,35
 32   PRINT 9017, RMAXP,ITERP
 9017 FORMAT(//77H *****  DIVERGENCE IN PSI-INNER-ITERATIONS. PROBLEM AB
     1ANDONED. MAX RESIDUAL =E15.6,8H AT ITER,I6 )
      MP=NP=1
      PRINT 9009
 9009 FORMAT(/ 20X,20HSTREAM FUNCTION, PSI )
      CALL PRTMAT(PSI)
      PRINT 9050
 9050 FORMAT(/ 20X,12HVORTICITY, W )
      CALL PRTMAT(W)
      GO TO 70
C   TEST PSI INNER-ITERATIONS FOR CONVERGENCE.
 35   IF(RMAXP.LE. TOLP) 40,45
 40   PRINT 915, ITERP,TOLP,RMAXP,ITER
 915  FORMAT( 26H *****  AT INNER-ITERATION,I6,10H TOLERANCE,E10.1,
     1 34H SATISFIED WITH MAXIMUM RESIDUAL =,E15.6,10H  FOR PSI(,I5,1H))
C   WEIGHT STREAM FUNCTION IN INTERIOR.
      DO 402 J=2,NB1
      DO 402 I=2,M1
 402  PSI(I,J)=XI*PSISAV(I,J)+XI1*PSI(I,J)
      DO 404 J=NB,N1
```

```
      DO 403 I=2,MSL1
403   PSI(I,J)=XI*PSISAV(I,J)+XI1*PSI(I,J)
      DO 404 I=MSRP1,M1
404   PSI(I,J)=XI*PSISAV(I,J)+XI1*PSI(I,J)
      IF(ITOL.LT. TOLTEST) 50,405
405   ITOL=0
      IF(ITER .EQ.1) GO TO 50
      RMAX1=0.0
C
C     COMPUTE OUTER-ITERATION RESIDUALS FOR STREAM FUNCTION.
      DO 415 J=2,NB1
      DO 415 I=2,M1
      RES=ABSF(PSI(I,J)-PSISAV(I,J))
      IF(RES .GT. RMAX1) 41,415
41    RMAX1=RES
415   CONTINUE
      DO 43 J=NB,N1
      DO 422 I=2,MSL1
      RES=ABSF(PSI(I,J)-PSISAV(I,J))
      IF(RES .GT. RMAX1) 42,422
42    RMAX1=RES
422   CONTINUE
      DO 43 I=MSRP1,M1
      RES=ABSF(PSI(I,J)-PSISAV(I,J))
      IF(RES .GT. RMAX1) 425,43
425   RMAX1=RES
43    CONTINUE
      RMAX=RMAXPSV=RMAX1
C
C     TEST PSI OUTER-ITERATIONS FOR DIVERGENCE.
      IF(RMAX .GT.1.E+5 ) 432,435
432   PRINT 9432, ITER,RMAX
9432  FORMAT(//56H *****  DIVERGENCE IN STREAM FUNCTION AT OUTER-ITERATI
     1ON,I6,20H  MAXIMUM RESIDUAL =,E12.4 )
      MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
      GO TO 70
C     TEST PSI OUTER-ITERATIONS FOR CONVERGENCE.
C     IF CONVERGENCE, GO TO COMPUTE AND TEST VORTICITY.
435   PCONV=0
      IF(RMAX .LE. TOLP) 440,448
440   PRINT 9440, ITER,TOLP,RMAX
9440  FORMAT(    26H    ***  AT OUTER-ITERATION,I6,14H PSI-TOLERANCE,E10.1
     1 ,34H SATISFIED WITH MAXIMUM RESIDUAL =,E15.6 /)
      PCONV=1
      GO TO 50
C
C     FOR VORTICITY COMPUTE OUTER-ITERATION RESIDUALS EVERYWHERE EXCEPT
C     LEFT AND RIGHT BOUNDARIES.
4402  RMAX1=0.0
      DO 441 J=1,NB
      DO 441 I=2,M1
      RES=ABSF(W(I,J)-WSAV(I,J))
```

```
      IF(RES .GT. RMAX1) 4405,441
 4405 RMAX1=RES
 441  CONTINUE
      DO 443 J=NBP1,N
      DO 442 I=2,MSL
      RES=ABSF(W(I,J)-WSAV(I,J))
      IF(RES .GT. RMAX1) 4415,442
 4415 RMAX1=RES
 442  CONTINUE
      DO 443 I=MSR,M1
      RES=ABSF(W(I,J)-WSAV(I,J))
      IF(RES .GT. RMAX1) 4425,443
 4425 RMAX1=RES
 443  CONTINUE
      RMAX=RMAX1
C
C   TEST VORTICITY OUTER-ITERATIONS FOR DIVERGENCE.
      IF(RMAX .GT.1.E+5) 4432,4435
 4432 PRINT 9532, ITER,RMAX
 9532 FORMAT(//50H *****  DIVERGENCE IN VORTICITY AT OUTER-ITERATION,I6,
     1 20H  MAXIMUM RESIDUAL =,E12.4 )
      MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
      GO TO 70
C   TEST VORTICITY OUTER-ITERATIONS FOR CONVERGENCE.
C   IF CONVERGENCE, AND IF PSI HAS CONVERGED, SOLUTION OBTAINED.
 4435 IF(RMAX .LE. TOLW) 4439,445
 4439 PRINT 9443, ITER,TOLW,RMAX
 9443 FORMAT(    26H    ***  AT OUTER-ITERATION,I6,12H W-TOLERANCE,E10.1
     1 ,34H SATISFIED WITH MAXIMUM RESIDUAL =,E15.6 /)
      IF(PCONV) 444,48
 444  MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
C   SAVE SOLUTION ON TAPE, IF REQUIRED.
 4441 IF(TAPEUSE) 4442,70
 4442 REWIND 5
      DO 4444 J=1,N
 4444 WRITE (5) (PSI(I,J),I=1,M)
      DO 4445 J=1,N
 4445 WRITE (5) (W(I,J),I=1,M)
      GO TO 70
C   TEST IF MAXIMUM OUTER ITERATIONS EXCEEDED.
 445  PRINT 9445, ITER,RMAX
 9445 FORMAT(26H    ***  AT OUTER-ITERATION,I6,34H  MAXIMUM RESIDUAL FOR
     1VORTICITY =,E15.6 /)
      IF(ITER .GE. ITERMAX) 447,48
 447  PRINT 913
 913  FORMAT(//65H *****  MAXIMUM NUMBER OF OUTER-ITERATIONS USED. ABAND
     1ON PROBLEM. )
      MP=NP=1
```

```
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
      GO TO 4441
C
 448  PRINT 9448, ITER,RMAXPSV
 9448 FORMAT(26H    ***  AT OUTER-ITERATION,I6,40H  MAXIMUM RESIDUAL FOR
     1STREAM FUNCTION =,E15.6 /)
      GO TO 50
C    TEST IF MAXIMUM PSI INNER-ITERATIONS EXCEEDED.
 45   IF(ITERP.GE. ITERMAXP) 47,106
 47   PRINT 9013, RMAXP,ITERP
 9013 FORMAT(// 88H *****  MAXIMUM NUMBER OF INNER-ITERATIONS USED FOR S
     1TREAM FUNCTION.  MAXIMUM RESIDUAL =E12.4,8H AT ITER,I6 )
      MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
      GO TO 70
C
 48   PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
      PRINT 9480
 9480 FORMAT(///)
      GO TO 10
C
C    BLOCK TO COMPUTE CW-COEFFICIENTS FOR VORTICITY.
 4800 A=(PSI(I,J+1)-PSI(I,J-1))*CW10*CW12
      B=((PSI(I+1,J)-PSI(I-1,J))*CW10+3.+CW13)*CW12
      IF(A   .GE. 0.) 4851,4855
 4851 IF(B   .GE. 0.) 4852,4853
 4852 CW0=4.0+A+B
      CW1=1.0
      CW2=1.-.5*CW12+B
      CW3=1.+A
      CW4=1.+.5*CW12
      GO TO 4860
 4853 CW0=4.0+A-B
      CW1=1.0
      CW2=1.-.5*CW12
      CW3=1.+A
      CW4=1.+.5*CW12-B
      GO TO 4860
 4855 IF(B   .GE. 0.) 4856,4857
 4856 CW0=4.0-A+B
      CW1=1.0-A
      CW2=1.-.5*CW12+B
      CW3=1.0
      CW4=1.+.5*CW12
      GO TO 4860
 4857 CW0=4.0-A-B
      CW1=1.0-A
```

```
      CW2=1.-.5*CW12
      CW3=1.0
      CW4=1.+.5*CW12-B
C
 4860 CW0=OMEGAW/CW0
      GO TO (525,53,54,639,6403,648), RET
C
 50   RMAXW=1.E91
C   SAVE VORTICITY OF PREVIOUS OUTER-ITERATION.
      DO 502 J=1,N
      DO 502 I=1,M
 502  WSAV(I,J)=W(I,J)
C   COMPUTE VORTICITY ON TOP AND BOTTOM BOUNDARIES FOR THIS OUTER-ITER.
C   THE LINE AB.
      DO 5021 I=2,MSL1
 5021 W(I,N)= CW5*(1.-PSI(I,N1))
C   THE LINE CD.
      DO 5022 I=MSLP1,MSR1
 5022 W(I,NB)=CW8*(1.-PSI(I,NB1))
C   THE LINE EF.
      DO 5023 I=MSRP1,M1
 5023 W(I,N)= CW5*(1.-PSI(I,N1))
C   THE LINES CB AND ED.
      DO 5024 J=NBP1,N1
      KK=J-1
      CW14=1./(KK*KK)
      W(MSR,J)=(1.-PSI(MSRP1,J))*CW9*CW14
 5024 W(MSL,J)=(1.-PSI(MSL1,J))*CW9*CW14
C   THE LINE HG.
      DO 5025 I=2,M1
 5025 W(I,1)=W(I,2)
C   THE CORNER POINTS C AND D.
      W(MSR,NB)=CW6*(2.+CW7-PSI(MSRP1,NB)-(1.+CW7)*PSI(MSR,NB1))
      W(MSL,NB)=CW6*(2.+CW7-PSI(MSL1,NB)-(1.+CW7)*PSI(MSL,NB1))
C
C   BEGIN VORTICITY INNER-ITERATIONS.
      ITERW=0
 505  ITOLW=0
 506  ITERW=ITERW+1
      ITOLW=ITOLW+1
      IF(ITOLW .LT. TOLTESTW) 507,63
C   SWEEP VORTICITY IN INTERIOR.
C   THE RECTANGULAR REGION BELOW THE STEP.
 507  RET=1
      DO 525 J=2,NB1
      CW12=1./(J-1)
      CW13=CW11*CW12*CW12
      DO 525 I=2,M1
      GO TO 4800
 525  W(I,J)=CW00*W(I,J)+CW0*(CW1*W(I+1,J)+CW2*W(I,J+1)+CW3*W(I-1,J)
     1 + CW4*W(I,J-1))
C   THE RECTANGULAR REGION ABOVE THE STEP ON THE LEFT.
      DO 55 J=NB,N1
      RET=2
      CW12=1./(J-1)
      CW13=CW11*CW12*CW12
```

```
      DO 53 I=2,MSL1
      GO TO 4800
 53   W(I,J)=CW00*W(I,J)+CW0*(CW1*W(I+1,J)+CW2*W(I,J+1)+CW3*W(I-1,J)
     1 + CW4*W(I,J-1))
C    THE RECTANGULAR REGION ABOVE THE STEP ON THE RIGHT.
      RET=3
      DO 54 I=MSRP1,M1
      GO TO 4800
 54   W(I,J)=CW00*W(I,J)+CW0*(CW1*W(I+1,J)+CW2*W(I,J+1)+CW3*W(I-1,J)
     1 + CW4*W(I,J-1))
 55   CONTINUE
      GO TO 506
C
C    SWEEP VORTICITY IN INTERIOR AND COMPUTE RESIDUALS.
 63   RMAX1W=0.0
C    THE RECTANGULAR REGION BELOW THE STEP.
      RET=4
      DO 640 J=2,NB1
      CW12=1./(J-1)
      CW13=CW11*CW12*CW12
      DO 640 I=2,M1
      GO TO 4800
 639  WOLD=W(I,J)
      WNEW  =CW00*W(I,J)+CW0*(CW1*W(I+1,J)+CW2*W(I,J+1)+CW3*W(I-1,J)
     1 + CW4*W(I,J-1))
      W(I,J)=WNEW
      RES=ABSF(WNEW-WOLD)
      IF(RES .GT.RMAX1W) 6395,640
 6395 RMAX1W=RES
 640  CONTINUE
C    THE RECTANGULAR REGION ABOVE THE STEP ON THE LEFT.
      DO 651 J=NB,N1
      RET=5
      CW12=1./(J-1)
      CW13=CW11*CW12*CW12
      DO 641 I=2,MSL1
      GO TO 4800
 6403 WOLD=W(I,J)
      WNEW  =CW00*W(I,J)+CW0*(CW1*W(I+1,J)+CW2*W(I,J+1)+CW3*W(I-1,J)
     1 + CW4*W(I,J-1))
      W(I,J)=WNEW
      RES=ABSF(WNEW-WOLD)
      IF(RES .GT.RMAX1W) 6405,641
 6405 RMAX1W=RES
 641  CONTINUE
C    THE RECTANGULAR REGION ABOVE THE STEP ON THE RIGHT.
      RET=6
      DO 650 I=MSRP1,M1
      GO TO 4800
 648  WOLD=W(I,J)
      WNEW  =CW00*W(I,J)+CW0*(CW1*W(I+1,J)+CW2*W(I,J+1)+CW3*W(I-1,J)
     1 + CW4*W(I,J-1))
      W(I,J)=WNEW
      RES=ABSF(WNEW-WOLD)
      IF(RES .GT.RMAX1W) 649,650
 649  RMAX1W=RES
```

```
  650  CONTINUE
  651  CONTINUE
       RMAXW=RMAX1W
C
C   TEST VORTICITY INNER-ITERATIONS FOR DIVERGENCE.
       IF(RMAXW.GT. 1.E+5 ) 665,666
  665  PRINT 9665, RMAXW,ITERW
 9665  FORMAT(// 55H *****  DIVERGENCE IN W-ONLY ITERATIONS. MAX RESIDUAL
      1 =,E12.4,8H AT ITER,I6 )
       MP=NP=1
       PRINT 9009
       CALL PRTMAT(PSI)
       PRINT 9050
       CALL PRTMAT(W)
       GO TO 70
C   TEST VORTICITY INNER-ITERATIONS FOR CONVERGENCE.
C   IF CONVERGENCE, THEN WEIGHT AND GO TO TEST OUTER-ITERATIONS.
  666  IF(RMAXW .LE. TOLW) 67,675
  67   PRINT 9067, ITERW,TOLW,RMAXW,ITER
 9067  FORMAT( 26H *****  AT INNER-ITERATION,I6,10H TOLERANCE,E10.1,
      1 34H SATISFIED WITH MAXIMUM RESIDUAL =,E15.6,8H  FOR W(,I5,1H) /)
C   WEIGHT VORTICITY EVERYWHERE EXCEPT LEFT AND RIGHT BOUNDARIES.
       DO 6715 J=1,NB
       DO 6715 I=2,M1
 6715  W(I,J)=DELTA*WSAV(I,J)+DELTA1*W(I,J)
       DO 672 J=NBP1,N
       DO 6717 I=2,MSL
 6717  W(I,J)=DELTA*WSAV(I,J)+DELTA1*W(I,J)
       DO 672 I=MSR,M1
  672  W(I,J)=DELTA*WSAV(I,J)+DELTA1*W(I,J)
       IF(ITOL .EQ. 0) 4402,10
C   TEST IF MAXIMUM VORTICITY INNER-ITERATIONS EXCEEDED.
  675  IF(ITERW .GE.ITERMAXW) 677,505
  677  PRINT 9677 , RMAXW,ITERW
 9677  FORMAT(// 80H *****  MAXIMUM NUMBER OF ITERATIONS USED FOR W-INNER
      1-ITERATIONS. MAX RESIDUAL =,E12.4,8H AT ITER,I6 )
       MP=NP=1
       PRINT 9009
       CALL PRTMAT(PSI)
       PRINT 9050
       CALL PRTMAT(W)
C
C   END OF MAIN LOOP
  70   CONTINUE
       STOP
       END
```